// File: checker.sol



//SPDX-License-Identifier: Unlicense
pragma solidity ^0.8.0;

import "Ownable.sol";

interface IBentoBox {
  struct StrategyData {
    uint64 strategyStartDate;
    uint64 targetPercentage;
    uint128 balance;
  }

  struct Rebase {
    uint128 elastic;
    uint128 base;
  }

  function totals(address token) external view returns (Rebase memory);
  function strategyData(address token) external view returns (StrategyData memory);
  function harvest(address token, bool rebalance, uint256 maxChange) external;
}

contract UtilizationChecker is Ownable {
  IBentoBox constant BentoBox = IBentoBox(0xF5BCE5077908a1b7370B9ae04AdC565EBd643966);

  uint256 constant UTILIZATION_PRECISION = 1e6;

  struct Strategy {
    address token;
    uint256 maxUtilization;
    uint256 minUtilization;
    bool enabled;
  }

  Strategy[] public strategies;

  function addStrategies(address[] calldata tokens, uint256[] calldata maxUtilizations, uint256[] calldata minUtilizations) external onlyOwner {
    for(uint256 i = 0; i < tokens.length; i++) {
      strategies.push(Strategy(tokens[i], maxUtilizations[i] * UTILIZATION_PRECISION, minUtilizations[i] * UTILIZATION_PRECISION, true));
    }
  }

  function setStrategies(uint256[] calldata indexes, uint256[] calldata maxUtilizations, uint256[] calldata minUtilizations, bool[] calldata enableds) external onlyOwner {
    for(uint256 i = 0; i < indexes.length; i++) {
      strategies[indexes[i]] = Strategy(strategies[indexes[i]].token, maxUtilizations[i] * UTILIZATION_PRECISION, minUtilizations[i] * UTILIZATION_PRECISION, enableds[i]);
    }
  }

  function checker() external view returns (bool canExec, bytes memory execPayload) {
    for(uint256 i = 0; i < strategies.length; i++) {
      Strategy memory strategy = strategies[i];

      if(strategy.enabled == false) {
        continue;
      }

      IBentoBox.Rebase memory totals = BentoBox.totals(strategy.token);
      IBentoBox.StrategyData memory strategyData = BentoBox.strategyData(strategies[i].token);

      uint256 currentPercentage = (strategyData.balance * UTILIZATION_PRECISION / totals.elastic) * 100;

      if(currentPercentage >= strategy.maxUtilization || currentPercentage <= strategy.minUtilization) {
        canExec = true;
        execPayload = abi.encodeWithSelector(
            IBentoBox.harvest.selector,
            strategy.token,
            true,
            0
        );

        break;
      }
    }
  }
}
// File: Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/Ownable.sol)

pragma solidity ^0.8.0;

import "Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}